/*
Seth Goldman
Proposal title: The Victim Paradox: How Portrayals of Suffering Affect Attitudes about Gay People and Gay Rights


HYPOTHESES

Stated-Hyp1: "The victim narrative should produce the most sympathy because it
portrays the most suffering, followed by the narrative showing a modest amount of suffering, and, finally, the narrative that does not portray any suffering" (p. 6)
	// variable that best maps onto sympathy is compassion.
	
	Test-Hyp1: Compassion will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	
	Test-Hyp2: Compassion will be higher in medium suffering vs. low suffering condition.
	
	Test-Hyp3: Compassion will be higher in low suffering vs. no suffering condition.
	
	
Stated-Hyp2: "I expect to find the same pattern of effects with regard to outgroup attitudes... If the victim narrative produces the most sympathy, then it should also produce the largest increases in positive feelings about the outgroup." (p. 6)
[strongest effect for victim narrative, then modest suffering, then no suffering.]

	Test-Hyp4: Warmth towards LGB individuals will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	
	Test-Hyp5: Warmth towards LGB individuals will be higher in medium suffering vs. low suffering condition.
	
	Test-Hyp6: Warmth towards LGB individuals will be higher in low suffering vs. no suffering condition.
	

Stated-Hyp3: "The victim narrative's effects on feelings about the outgroup should, in turn, lead to greater support for public policies benefiting the outgroup.... In addition to this path of influence, sympathy for the victim may directly influence policy support" (p. 6)

	Test-Hyp7: Support for gay marriage will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	
	Test-Hyp8: Support for gay marriage will be higher in medium suffering vs. low suffering condition.
	
	Test-Hyp9: Support for gay marriage will be higher in low suffering vs. no suffering condition.

	Test-Hyp10: Higher support for gay marriage in victim narrative (highest suffering) vs. medium suffering condition will be mediated by warmth towards LGB individuals.
	
	Test-Hyp11: Higher support for gay marriage in medium suffering vs. low suffering condition will be mediated by warmth towards LGB individuals.
	
	Test-Hyp12: Higher support for gay marriage in low suffering vs. no suffering condition will be mediated by warmth towards LGB individuals.
	
	Test-Hyp13: Higher support for gay marriage in victim narrative (highest suffering) vs. medium suffering condition will be mediated by compassion.
	
	Test-Hyp14: Higher support for gay marriage in medium suffering vs. low suffering condition will be mediated by compassion.
	
	Test-Hyp15: Higher support for gay marriage in low suffering vs. no suffering condition will be mediated by compassion.
	
		
Stated-Hyp4: The victim narrative will produce pity. (p. 7)

	Test-Hyp16: Pity will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	Test-Hyp17: Pity will be higher in medium suffering vs. low suffering condition.
	Test-Hyp18: Pity will be higher in low suffering vs. no suffering condition.
	
	
Stated-Hyp5: The victim narrative will produce "paternalistic prejudice". (p. 7)

	Test-Hyp19: Paternalism will be higher in victim narrative (highest suffering) vs. medium suffering condition.	
	Test-Hyp20: Paternalism will be higher in medium suffering vs. low suffering condition.	
	Test-Hyp21: Paternalism will be higher in low suffering vs. no suffering condition.
	
	
Stated-Hyp5: "Exposure to situations that reinforce victimhood due to being gay should increase internalized stigma." (p. 7) 
[From the rest of the discussion in this paragraph, it seems that this prediction pertains to LGBT respondents only.]

	Test-Hyp22: Internalized stigma will be higher in victim narrative vs. control condition.
		
	
********************************************************************************
NOTES

- For stated hypotheses 3 and 4, we are assuming the relationships expected in 
prior hypotheses to hold (i.e. strongest effect for victim narrative, then 
modest suffering, then no suffering.)	

- Paternalistic prejudice items are listed on page 29 of the proposal.

- Note: hero narrative is medium suffering condition (p. 5).

- For test hypotheses 10-12, there is no measure of sympathy in the survey, but 
there is a measure of compassion which probably is what the author meant by 
sympathy. However, sympathy is measures for the protagonist only, not all LGB 
individuals.  

- For stated hypothesis 5, the author wants to measure the effect of the victim 
narrative only. I am therefore only comparing victim narrative with control 
condition.

- For test hypotheses 12 and 15, the effect is mediated as expected but the main 
effect is in the wrong direction. 
*/

clear all
use "Goldman507.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS
	tab XTESS163
	recode XTESS163 (1 5=0 "0 control") (4=1 "1 low suffering") (4 6 =2 "2 medium suffering") (3 7=3 "3 victim narrative/ highest suffering"), gen(treatment)
	tab treatment

	* victim narrative vs. medium suffering
	recode treatment (3=1) (2=0) (*=.), gen(victim_medium)
	tab victim_medium, mis
	
	* median vs. low suffering
	recode treatment (2=1) (1=0) (*=.), gen(medium_low)
	tab medium_low, mis	
	
	* low vs. no suffering
	recode treatment (1=1) (0=0) (*=.), gen(low_none)
	tab low_none, mis	
		
	* victim narrative vs. no suffering
	recode treatment (0=0) (3=1) (*=.), gen(victim_none)
	tab victim_none, mis
		
	
* OUTCOME MEASURES
	
* feelings of compassion, warmth, and pity towards protagonist
	* label 
	lab def feeling 1 "not at all" 7 "extremely"

	* recode missing
foreach var in Q5_A_ORIGINAL Q5_B_ORIGINAL Q5_C_ORIGINAL Q5_A_UPDATED Q5_B_UPDATED Q5_C_UPDATED {
	tab `var'
	replace `var'=. if `var'==-1
	lab val `var' feeling
	tab `var', mis
}

	* respondents were either asked the original or updated version, but the questions are the same; combining into single measures
	clonevar compassion =Q5_A_ORIGINAL
	replace compassion = Q5_A_UPDATED if compassion==.
	tab compassion, missing
	
	clonevar warmth =Q5_B_ORIGINAL
	replace warmth = Q5_B_UPDATED if warmth==.
	tab warmth, missing
	
	clonevar pity =Q5_C_ORIGINAL
	replace pity = Q5_C_UPDATED if pity==.
	tab pity, mis
	
* policy support 
	* support for gay marriage	
	tab Q16
	replace Q16=. if Q16==-1
	gen support_gaymarriage= Q16
	tab support_gaymarriage
	
* outgroup attitude 
	/*How do you feel about gays and lesbians?  On the feeling thermometer below, 0 means you feel extremely cold and 10 means you feel extremely warm.  Please use the thermometer to say how you feel about gays and lesbians. 
	*/
	tab Q17
	replace Q17=. if Q17==-1
	gen warmth_lgb = Q17

	
* paternalistic prejucide (see proposal p. 29)
	/*
1.	If given a choice, most gay people would not change their sexual orientation. 
2.	I feel sorry for gay and lesbian people.
3.	Gay people are in desperate need of society's acceptance and support. 

(more agreement shows greater paternalism; all items are reverse coded)

*/
	
	* recode missing 
	foreach var in Q18_1 Q18_2 Q18_3 {
		replace `var'=. if `var'==-1
		tab `var'
	}
	
	* recode so higher values indicate higher prejudice
	foreach var in Q18_1 Q18_2 Q18_3 {
		gen `var'R=	6-`var'
		tab `var'R
	}	
	

	* create summated scale
	gen paternalism =Q18_1R + Q18_2R +Q18_3R
	tab paternalism
	
* internalized stigma (only asked of LGB respondents)
	/*
1.	I would have had better chances in life if I had not been LGB--REVERSE
2.	My experiences as a LGB person have made me a better person. 
3.	I sometimes feel alone because I am LGB.--REVERSE
4.	I am proud to be LGB. 
5.	When it comes to disclosing your sexual orientation, you can't be too careful.--REVERSE
	*/

	* recode missing
	foreach var in Q19_1 Q19_2 Q19_3 Q19_4 Q19_5 {
		replace `var'=. if `var'==-1
		tab `var'
	}
	
	* recode so higher values indicate higher internalized stigma 
	foreach var in Q19_1 Q19_3 Q19_5 {
		gen `var'R=	6-`var'
		tab `var'R
	}	

	* create summated index
	gen internalizedstigma= Q19_1R+ Q19_2+ Q19_3R+ Q19_4 +Q19_5R
	tab internalizedstigma
	
	
********************************************************************************

* ANALYSIS
	

*Test-Hyp1: Compassion will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	reg compassion i.victim_medium 
	// do not reject. 0.000
	tess 1.victim_medium +, init(Goldman507) bonf(5)
	
	
*Test-Hyp2: Compassion will be higher in medium suffering vs. low suffering condition.
	reg compassion i.medium_low 
	// do not reject. 0.000
	tess 1.medium_low +, bonf(5)
	
	
*Test-Hyp3: Compassion will be higher in low suffering vs. no suffering condition.
	reg compassion i.low_none 
	// do not reject. 0.000	
	tess 1.low_none +, bonf(5)
	
	
*Test-Hyp4: Warmth towards LGB individuals will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	reg warmth_lgb i.victim_medium 
	// reject. 0.378
	tess 1.victim_medium +, bonf(5)
	
	
*Test-Hyp5: Warmth towards LGB individuals will be higher in medium suffering vs. low suffering condition.
	reg warmth_lgb i.medium_low 
	// reject. 0.697
	tess 1.medium_low +, bonf(5)
	
	
*Test-Hyp6: Warmth towards LGB individuals will be higher in low suffering vs. no suffering condition.
	reg warmth_lgb i.low_none 
	// reject. 0.887		
	tess 1.low_none +, bonf(5)
	
	
*Test-Hyp7: Support for gay marriage will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	reg support_gaymarriage i.victim_medium 
	// reject. 0.381	
	tess 1.victim_medium +, bonf(5)
	
	
*Test-Hyp8: Support for gay marriage will be higher in medium suffering vs. low suffering condition.
	reg support_gaymarriage i.medium_low
	// do not reject. 0.000
	tess 1.medium_low +, bonf(5)
	
	
*Test-Hyp9: Support for gay marriage will be higher in low suffering vs. no suffering condition.
	reg support_gaymarriage i.low_none
	// reject. 0.000W
	tess 1.low_none +, bonf(5)
	
	
*Test-Hyp10: Higher support for gay marriage in victim narrative (highest suffering) vs. medium suffering condition will be mediated by warmth towards LGB individuals.
	est sto clear
	reg support_gaymarriage victim_medium 
	est sto m1
	reg support_gaymarriage victim_medium warmth_lgb
	est sto m2
	
	suest m1 m2
	test [m1_mean]victim_medium=[m2_mean]victim_medium

	tess, mediation bonf(2)
	// reject. 0.616
	
*Test-Hyp11: Higher support for gay marriage in medium suffering vs. low suffering condition will be mediated by warmth towards LGB individuals.
	est sto clear
	reg support_gaymarriage medium_low 
	est sto m1
	reg support_gaymarriage medium_low warmth_lgb
	est sto m2
	
	suest m1 m2
	test [m1_mean]medium_low=[m2_mean]medium_low

	tess, mediation	bonf(2)
	// do not reject. 0.000
	
*Test-Hyp12: Higher support for gay marriage in low suffering vs. no suffering condition will be mediated by warmth towards LGB individuals.
	est sto clear
	reg support_gaymarriage low_none 
	est sto m1
	reg support_gaymarriage low_none warmth_lgb
	est sto m2
	
	suest m1 m2
	test [m1_mean]low_none=[m2_mean]low_none

	tess, mediation	bonf(2)
	// do not reject. 0.000
	
*Test-Hyp13: Higher support for gay marriage in victim narrative (highest suffering) vs. medium suffering condition will be mediated by compassion.
	
	est sto clear
	reg support_gaymarriage victim_medium 
	est sto m1
	reg support_gaymarriage victim_medium compassion
	est sto m2
	
	suest m1 m2
	test [m1_mean]victim_medium=[m2_mean]victim_medium

	tess, mediation	bonf(2)
	// do not reject. 0.028	
	
	
*Test-Hyp14: Higher support for gay marriage in medium suffering vs. low suffering condition will be mediated by compassion.
	est sto clear
	reg support_gaymarriage medium_low 
	est sto m1
	reg support_gaymarriage medium_low compassion
	est sto m2
	
	suest m1 m2
	test [m1_mean]medium_low=[m2_mean]medium_low

	tess, mediation	bonf(2)
	// do not reject. 0.000
		
*Test-Hyp15: Higher support for gay marriage in low suffering vs. no suffering condition will be mediated by compassion.
	est sto clear
	reg support_gaymarriage low_none 
	est sto m1
	reg support_gaymarriage low_none compassion
	est sto m2
	
	suest m1 m2
	test [m1_mean]low_none=[m2_mean]low_none
	tess, mediation bonf(2)
	// do not reject. 0.005	

*Test-Hyp16: Pity will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	reg pity i.victim_medium
	// do not reject. 0.000
	tess 1.victim_medium +, bonf(5)
	
*Test-Hyp17: Pity will be higher in medium suffering vs. low suffering condition.
	reg pity i.medium_low
	// do not reject. 0.000
	tess 1.medium_low +, bonf(5)
	
*Test-Hyp18: Pity will be higher in low suffering vs. no suffering condition.
	reg pity i.low_none
	// reject. 0.000W	
	tess 1.low_none +, bonf(5)	

*Test-Hyp19: Paternalism will be higher in victim narrative (highest suffering) vs. medium suffering condition.
	reg paternalism i.victim_medium
	// reject. 0.854
	tess 1.victim_medium +, bonf(5)
	
*Test-Hyp20: Paternalism will be higher in medium suffering vs. low suffering condition.
	reg paternalism i.medium_low
	// reject. 0.772
	tess 1.medium_low +, bonf(5)
	
*Test-Hyp21: Paternalism will be higher in low suffering vs. no suffering condition.
	reg paternalism i.low_none
	// reject. 0.833	
	tess 1.low_none +, bonf(5)
	
*Test-Hyp22: Internalized stigma will be higher in victim narrative vs. control condition.
	reg internalizedstigma i.victim_none
	// reject. 0.973	
	tess 1.victim_none +
